

Calcul de l'espérance et de la variance d'une variable suivant une loi exponentielle

On considère X une variable aléatoire continue suivant la loi exponentielle de paramètre a . Nous allons calculer son espérance mathématique et sa variance.

Commençons par rappeler la densité de probabilité de X . C'est la fonction f définie sur $[0; +\infty[$ par :

$$f : x \mapsto a e^{-ax}$$

```
> f(x) := a*exp(-a*x);
```

```
f(x) := a*EXP((-a)*x);
```

```
> assume(a>0)$
```

Nous savons que l'espérance de X est le réel, noté $E(X)$ égal à $\int_0^{+\infty} x f(x) dx$. Nous allons utiliser la méthode d'intégration par parties pour calculer cette intégrale. Nous avons énoncé en cours la propriété.

Le logiciel MAXIMA ne sait pas *mettre en scène* l'intégration par parties, aussi, allons-nous définir au préalable la commande qui présentera le calcul.

```
> ipp(u,v,x) := block([U],
  U:integrate(u,x),
  'integrate(u*v,x)=U*v-'integrate(U*diff(v,x),x)
)$
```

Vous pouvez constater que cela correspond à la formule donnée en cours.

Cette commande nous donnera une primitive de la fonction à intégrer $u v$, il faudra ensuite calculer l'intégrale correspondante sur l'intervalle $[0; +\infty[$.

```
> ipp(f(x),x,x);
```

$$a \int x e^{-ax} dx = \int e^{-ax} dx - x e^{-ax}$$

Il reste à calculer la variation de cette primitive entre 0 et $+\infty$.

```
> integrate(exp(-a*x),x,0,inf)+limit(x*exp(-a*x),x,inf)-limit(x*exp(-a*x),x,0);
```

$$\frac{1}{a}$$

Nous retrouvons donc bien que l'espérance de X est égale à $\frac{1}{a}$.

Remarque – MAXIMA sait calculer directement l'intégrale recherchée.

```
> 'integrate(x*f(x),x,0,inf)=integrate(x*f(x),x,0,inf);
```

$$a \int_0^{\infty} x e^{-ax} dx = \frac{1}{a}$$

Puisque vous avez compris le principe (!!), je vous propose de la même manière de calculer la variance de X .

Rappel : $Var(X) = \int_0^{+\infty} x^2 f(x) dx - (E(X))^2$.